Title: ISBLANK function in Excel to check if cell is blank
Headings: Excel ISBLANK function to check if cell is empty or not
Excel ISBLANK function to check if cell is empty or not
Excel ISBLANK function
ISBLANK (value)
=ISBLANK (A2)
=NOT (ISBLANK (A2))
ISBLANK in Excel - things to remember
How to use ISBLANK in Excel
Excel formula: if cell is blank then
=IF (ISBLANK (B2), "Open", "Completed")
=IF (Sheet3!B2<>"",Sheet3!B2,"")
=IF (B2="", "Open", "Completed")
Excel formula: if cell is not blank then
=IF (NOT (ISBLANK (B2)), "Completed", "")
=IF (ISBLANK (B2), "", "Completed")
=IF (B2<>"", "Completed", "")
If cell is blank, then leave blank
=IF (ISBLANK (B2), "", B2*10%)
=IF (B2="", "", B2*10%)
If any cell in range is blank, then do something
COUNTBLANK (range)>0
SUMPRODUCT (-- (range=""))>0
=IF (COUNTBLANK (B2:D2)>0,"Open", "")
=IF (COUNTIF (B2:D2,"")>0, "Open", "")
=IF (SUMPRODUCT (-- (B2:D2=""))>0, "Open", "")
If all cells in range are blank, then do something
=IF (COUNTA (B2:D2)>0, "", "Not Started")
=IF (COUNTIF (B2:D2,"<>")>0, "", "Not Started")
=IF (SUMPRODUCT (-- (B2:D2<>""))>0, "", "Not Started")
=IF (AND (ISBLANK (B2:D2)), "Not Started", "")
Excel formula: if cell is not blank, then sum
If not blank then sum
=SUMIF (B2:B6, "<>", C2:C6)
If blank then sum
=SUMIF (B2:B6, "", C2:C6)
Sum if all cells in range are not blank
=IF (COUNTBLANK (B2:B6)=0, SUM (B2:B6), "")
=IF (OR (ISBLANK (B2:B6)), "", SUM (B2:B6))
Excel formula: count if cell is not blank
=COUNTA (B2:B6)
=COUNTIF (B2:B6,"<>")
=COUNTBLANK (B2:B6)
Excel ISBLANK not working
Treat zero-length strings as blanks
=IF (A2="", "blank", "not blank")
=IF (LEN (A2)=0, "blank", "not blank")
Remove or ignore extra spaces
=IF (OR (A2="", LEN (TRIM (A2))=0), "blank", "not blank")
=IF (OR (A2="", A2=CHAR (160)), "blank", "not blank")
Available downloads
You may also be interested in

Content: =ISBLANK (A2)
To check if A2 is not empty, use ISBLANK together with the NOT function, which returns the reversed logical value, i.e. TRUE for non-blanks and FALSE for blanks. =NOT (ISBLANK (A2))
Copy the formulas down to a few more cells and you will get this result: ISBLANK in Excel - things to remember
The main point you should keep in mind is that the Excel ISBLANK function identifies truly empty cells, i.e. cells that contain absolutely nothing: no spaces, no tabs, no carriage returns, nothing that only appears blank in a view. For a cell that looks blank, but in fact is not, an ISBLANK formula returns FALSE. This behavior occurs if a cell contains any of the following: Formula that returns an empty string like IF (A1<>"", A1, ""). Zero-length string imported from an external database or resulted from a copy/paste operation. Spaces, apostrophes, non-breaking spaces ( ), linefeed or other non-printing characters. How to use ISBLANK in Excel
To gain more understanding of what the ISBLANK function is capable of, let's take a look at some practical examples.